

Microsoft Excel Calculating a Trend



Calculating a Trend

Use the TREND function to calculate a linear trend. This fits a straight line, using the method of least squares, to known dependent and independent variables and returns a new set of dependant variables. A trend is a statistical method of projecting for the future a set of outcomes based on data from the past. To fully understand the calculation method and interpret the results correctly you should consult the Excel documentation and also a publication dealing with statistical theory, search for the subject *linear regression*.

TREND(known_y's,known_x's,new_x's,const)

Where *Known_y's* and *Known_x's* are the known dependent and independent variables respectively. *New_x's* are the new independent variables for which you want TREND to return the corresponding dependent variables.

| B3 | | fx {=TREND(B2:D2,B1:D1,B1:F1)} | | | | |
|----|---------|--------------------------------|-------|-------|-------|-------|
| | A | B | C | D | E | F |
| 1 | Year | 2003 | 2004 | 2005 | 2006 | 2007 |
| 2 | Actuals | 2,536 | 2,845 | 3,289 | | |
| 3 | Trend | 2,514 | 2,890 | 3,267 | 3,643 | 4,020 |

In the illustration, known data from the period 2003 to 2005 is being used to calculate

a trend for the period 2003 to 2007.

Single Cell Array Formulas

This is an example of where a formula has to be entered into a single cell as an array in order for it to calculate correctly. The formula does not return an array of values, rather the values that it is calculating need to be treated as an array. Even if you do not fully understand exactly how the formulas calculate you can nevertheless use the examples as a template to produce this type of calculation. In other words; "copy and paste" and change the cell references.

| | A | B | C | D |
|----|-------------|------------|-------|--------|
| 1 | Country | Product | Units | Value |
| 2 | UK | Soap | 1,789 | 81,460 |
| 3 | France | Toothpaste | 1,217 | 11,599 |
| 4 | Netherlands | Shave | 637 | 63,337 |
| 5 | Denmark | Laundry | 713 | 97,982 |
| 6 | Greece | Kitchen | 991 | 23,378 |
| 7 | Norway | Soap | 1,606 | 56,327 |
| 8 | Germany | Toothpaste | 46 | 11,101 |
| 9 | Spain | Shave | 888 | 87,187 |
| 10 | UK | Laundry | 1,244 | 92,504 |
| 11 | UK | Laundry | 1,537 | 82,624 |

The drawback to the function SUMIF is that it only accepts a single criterion value. To add numbers based on multiple conditions you either have to use DSUM or use the IF and SUM functions together entering them as an array formula into a single cell.

Select a single cell, type-in the formula and then press CTRL+SHIFT+ENTER. Do not enter them as normal formulas; they will give quite different results. If the entry fails, double-click the cell and try the array entry again.

Total the Units where the Product is Soap and the Country is UK:

=SUM(IF((B2:B11="Soap")*(A2:A11="UK"),C2:C11))

Total the Values where the Country is France or Spain:

=SUM(IF((A2:A11="France")+(A2:A11="Spain"),D2:D11))

Common problems

All examples assume that the first cell is A1.

Bad Negatives

Imported data sometimes has the negative numbers with the minus sign on the right. This is a common problem with data imported from SAP.

| Imported | Converted |
|----------|-----------|
| 43- | -43 |
| 34 | 34 |
| 335 | 335 |
| 45- | -45 |

The process of the formula is to test whether the first character on the right is a minus sign. If it is then place the minus sign in front of the number and replace the second instance of the minus sign with nothing. Finally, force the formula to return a numeric value.

Formula to convert the entries:

=VALUE(IF(RIGHT(A1,1)="-",SUBSTITUTE(RIGHT(A1,1)&A1,"-", "",2),A1))

Remainders

Areas to be tiled come in square metres but tiles come in packs of 12. Work out the number of packs to be ordered and the wastage. Floor or ceiling tiles.

| Sq m | Waste | Order |
|------|-------|-------|
| 23 | 11 | 2 |
| 2244 | 0 | 187 |
| 777 | 9 | 65 |
| 3556 | 4 | 297 |

The wastage is the remainder of the pack order that has not been used and the order is the square metres divided by 12 and rounded up to the nearest one if necessary.

Formula for the wastage:

=MOD(A1,12)

Formula for the order: **=IF(MOD(A1,12)=0,A1/12,CEILING(A1/12,1))**

League tables

Some people have to work harder at their formulas for Fantasy Football league than they do for their normal work. We have a table of names and total scores and we want to show a separate league table without having to copy and sort the table.

| Name | Score |
|-------|-------|
| Tom | 36 |
| Dick | 32 |
| Harry | 26 |
| Sid | 39 |
| Nancy | 23 |

| League table | | |
|--------------|----|-------|
| First | 39 | Sid |
| Second | 36 | Tom |
| Third | 32 | Dick |
| Forth | 26 | Harry |
| Fifth | 23 | Nancy |

Formula to calculate the highest score in column B is:

=LARGE(\$B\$1:\$B\$5,1)

To calculate the second highest etc.

=LARGE(\$B\$1:\$B\$5,2)

Carry on with the other formulas increasing the value by one. If the league table is long generate the sequential values required in the formula thus:

=LARGE(\$B\$1:\$B\$5,ROWS(A\$1:A1))

The formula to return the names in the last column of the league table:

=INDEX(\$A\$1:\$A\$5,MATCH(E1,\$B\$1:\$B\$5,0))

Find the index position of the matching league table score in column E in the scores listed in column B. Then return the item from the names listed in column A. NB* Does not work where the scores are tied.

Rounding

| | | |
|----------|--------|--------------|
| 5669.984 | 5669.9 | =TRUNC(A1,1) |
| 5669.984 | 5670 | =ROUND(A2,1) |
| 5669.984 | 5669 | =INT(A3) |

There are numerous rounding functions, all with slightly different effects.

